
=A1 Cell reference

='Sheet 2'!A1 Cell reference on another sheet/tab (Notice apostrophes, since tab name has a space)

=INDIRECT("A"&A1) Returns a cell reference specified by a string

=A1+2 Addition =A1-2 Subtraction =A1*2 Multiplication =A1/2 Division

=IF(A1=1,"Yes","No") If cell A1 equals 1, display the word “Yes”, if not display the word “No”

=if(A1<>"",B1-C1,"") Only performs the function if cell is not blank

=IF(A1=1,"Yes",IF(B1=1,"Yes","No")) Nested IF statement: If A1 = 1 display “Yes”. If A1 is not equal to 1… and if B1 = 1
display “Yes”. If A1 is not equal to 1 and B1 is not equal to 1, display “No”

=SORT(A1:B,1,true) Sort columns A and B, by column A (1st column), in ascending order

=SORT(A1:B,1,true,2,true) Sort by the 1st column, then by the 2nd

=FILTER(A1:A, B1:B=C1) Filter columns A and B, where column B equals the specified criteria

=FILTER(A1:C, B1:B="This", C1:C="And This") Filter with multiple conditions, where BOTH criteria must be true

=FILTER(A1:C, (B1:B="This")+(C1:C="Or That")) Filter with multiple conditions, where EITHER criteria can be true

=FILTER(A:B ,COUNTIF(C:C,A:A)) Filter based on a list. Filter A and B where A equals anything found in the list C

=FILTER(A1:1, A2:2=10) Filter horizontally  (Filter row 1 where row 2 equals 10)

=IFERROR(A1,"Error") If cell / formula has an error, display the text “Error”, if not, display the value in cell A1

=ARRAYFORMULA(A1:A) Reference to a column

=ARRAYFORMULA(A1:A*B1:B) Multiplies columns A and B

=ARRAYFORMULA(A1:A&" "&B1:B) Combines columns horizontally (Combine First + Last Name)

=ARRAYFORMULA(IF(A1:A>1,"Yes","No")) Using arrayformula to extend a function down the column

=UNIQUE(A1:A) Removes duplicates

=UNIQUE({A1:A12;B1:B12}) Combines columns vertically and removes duplicates

=COUNTUNIQUE(A1:A) Counts the unique values in a range

=COUNTIF(A1:A,B1) Counts the number of values in column A, that are equal to value that is in cell B1



=COUNTIF(A1:A,">100") Counts how many values are greater than 100 in column A

=COUNTIF(A1:A,"ExactText") Counts how many cells match the exact text specified

=COUNTIF(A1:A,"*PartialText*") Counts how many cells contain the text that is specified

=COUNTIF(A1:A,"<>") Counts non-empty cells

=COUNTIF(A1:A,">"&B1) Counts values in column A that are greater than the value that is in cell B1

=COUNTIFS(A2:A,">"&B8,A2:A,"<"&C8) Counts rows where BOTH criteria are true (Where values in column A are
greater than cell B8 and less than cell C8)

=TRANSPOSE(A1:B) Switches rows to columns, or vice versa

=SUM(A1:A) Sums the specified range

=SUMIF(A1:A,"Criteria",B1:B) Sums column B if the criteria is true in column A

=SUMIFS($E:$E, $B:$B, $A5, $A:$A, G$3) Sum column E, if BOTH criteria are true (A5 found in column B and G3
found in column A)

=AVERAGE(A1:A) Averages the specified range

=AVERAGEIF(A1:A,"Criteria",B1:B) Averages column B if the specified criteria is true in column A

=AVERAGEIFS(B1:B, A1:A, "Criteria", C1:C, "Criteria 2") Averages column B if BOTH criteria are true (“Criteria” found
in column A and “Criteria 2” found in column C)

=SPLIT(A1,"Criteria") Splits the contents of cells A1, by/at the specified criteria, into individual cells

=INDEX(SPLIT(A1, " "),1) Extracts Nth word

=INDEX(A:B,MATCH($C1,A:A,0)) Returns column A and B, where C1 is found in column A

=TRIM(A1) Removes leading and trailing spaces

=SUBSTITUTE(B3, " ", "") Removes all spaces

=LOWER(A1) Lowercase =UPPER(A1) Uppercase =PROPER(A1) Capitalize Words

=LEFT(C3,2) Extracts 2 characters from left of string =RIGHT(C3,2) Extracts 2 characters from right of string

=MIN(A:A) Finds the minimum of column A =MAX(A:A) Finds the maximum of column A

=ROUND(A1,2) Rounds number in cell A1 to 2 decimal places

=IMPORTRANGE("PasteGoogleSpreadSheetID/URL","'Tab'!A1:D") Imports data from another sheet/file

=QUERY(IMPORTRANGE("PasteGoogleSpreadSheetID/URL","'Tab'!A1:D"),"select Col1,Col3") Displays selected
columns from a specified imported range

=QUERY(Sheet1!A2:C500,"Select * where C contains 'Yes'", 0) Filters range where column contains a specified criteria



=QUERY(C2:C, "select C where C contains '"&I1&"' or C contains '"&L1&"' ") Filters
where column contains a criteria OR another criteria

=QUERY(C2:C, "select C where C contains '"&I1&"' and C contains '"&L1&"' ") Filters
where column contains a criteria AND another criteria

=QUERY($X:$Y," select X,sum(Y) where Y is not null group by X",0) Sums column, for each value in another column

=QUERY($X:$Y," select X,count(Y) where Y is not null group by X",0) Counts non-blank cells, for each value

=SPARKLINE(D3,{"charttype","bar";"max",E3;"color1","grey"}) Creates a bar chart within the cell, using cell D3 as the
data, and cell E3 as the maximum

=VALUE(REGEXREPLACE(A1,"[^[:digit:]]", "")) Extracts numbers from a string

=REGEXREPLACE(A1,"[^[:alpha:]]", "") Extracts text from a string

=REGEXREPLACE(A1,"[[:punct:]]", "") Removes punctuation from a string

=AND(A1=1,B1=1) Returns TRUE if BOTH criteria are true, and returns FALSE if not

=OR(A1=1,B1=1) Returns TRUE if EITHER criteria are true, and returns FALSE if not
______________________________________________________________________________________________________

Conditional formatting formulas

=COUNTIF(A:A,A1)>1 Highlights duplicates when entered in conditional formatting “Custom Formula Is”

=F5:F>=F$1 Format column F where criteria (Cell F1) is true

= $A1 = "Yes" Highlights entire row when criteria is true (Conditional formatting “Custom Formula Is”)
______________________________________________________________________________________________________

Seconds to Duration
=A1/(86400)

Minutes to duration
=A1/1440

Hours to duration
=A1/24

Duration to Seconds
=A1*86400

Duration to Minutes
=A1*1440

Duration to Hours
=A1*24

Mathematical Operators
Plus sign + ~  Addition
Minus sign - ~  Subtraction
Asterisk * ~  Multiplication
Forward Slash / ~  Division
Carrot ^ ~  Exponent

Comparison / Other operators
= (Equals) & (And) / (Concat)
> (Greater than) <> (Not equal to)
< (Less than)
>= (Greater than or equal to)
<= (Less than or equal to)

Keyboard Shortcuts
Ctrl + D Fill down Ctrl + C Copy
Ctrl + R Fill right iiwr Ctrl + V Paste
Ctrl + Z Undo Ctrl + A Select all
Ctrl + Y Redo Ctrl + F Find

Keyboard Shortcuts
Ctrl + Shift + V Paste values only
Arrow Keys Move from cell to cell
Shift + Arrow Keys Select multiple adjacent cells
Ctrl + Arrow Keys Move to end of data / sheet

